

Interception des erreurs et signalement à l'utilisateur

Les fonctions s'appuyant sur des valeurs saisies par les utilisateurs sont susceptibles de provoquer des erreurs qu'il est indispensable de capturer :

- afin de signaler l'erreur à l'utilisateur au travers d'un message clair
- et empêcher certains calculs qui pourraient être utilisés par ailleurs alors qu'ils sont basés sur des données erronées.

Ces fonctions permettent généralement la validation des données saisies par un utilisateur et sont souvent associées à la fonction **SI**, afin d'indiquer un message clair en retour.

Exemple 1 : avec RECHERCHEV

Les plages suivantes ont été nommées TAB_PERSONNEL et TAB_EMPLOI :

Personnel				Emploi	
Matricule	nom	prénom	code emploi	code emploi	Salaire de base
1010	Rigole	Jean	EPM	EMP	1 500,00 €
1020	Perret	Inès	TEC	TEC	1 600,00 €
1040	Golo	Henry	AGM	AGM	1 700,00 €
2000	Oposte	Fidel	TEC		

(erreur fiche 1010 : EPM n'existe pas...)

Pour une paie donnée, la saisie des matricules est utile pour rechercher le nom et le prénom :

B3									
=RECHERCHEV(A3;TAB_PERSONNEL;3;FAUX)&" "&RECHERCHEV(A3;TAB_PERSONNEL;2;FAUX)									
A	B	C	D	E	F	G	H	I	J
Paie du mois de Octobre 2014 (sans gestion d'erreur)									
Matricule	Nom	Salaire de base	Prime						
1000	=RECHERCHEV(A3;TAB_PERSONNEL;3;FAUX)&" "&RECHERCHEV(A3;TAB_PERSONNEL;2;FAUX)								
1020	Inès Perret	1 600,00 €	200,00 €						
1010	Jean Rigole	#N/A	100,00 €						
	#N/A	#N/A							
	totaux	#N/A	600,00 €						

et la recherche correspondant au code emploi de l'employé :

C3							
=RECHERCHEV(RECHERCHEV(A3;TAB_PERSONNEL;4;FAUX);TAB_EMPLOI;2;FAUX)							
A	B	C	D	E	F	G	H
Paie du mois de Octobre 2014 (sans gestion d'erreur)							
Matricule	Nom	Salaire de base	Prime				
1000	#N/A	=RECHERCHEV(RECHERCHEV(A3;TAB_PERSONNEL;4;FAUX);TAB_EMPLOI;2;FAUX)					
1020	Inès Perret	1 600,00 €	200,00 €				
1010	Jean Rigole	#N/A	100,00 €				
	#N/A	#N/A					
	totaux	#N/A	600,00 €				

A	B	C	D
Paie du mois de Octobre 2014 (sans gestion d'erreur)			
Matricule	Nom	Salaire de base	Prime
1000	#N/A	#N/A	300,00 €
1020	Inès Perret	1 600,00 €	200,00 €
1010	Jean Rigole	#N/A	100,00 €
	#N/A	#N/A	
	totaux	#N/A	600,00 €

Par chance, la somme des salaires provoque une erreur

Avec une gestion d'erreur :

1. Tout d'abord une valeur d'erreur a été décidée et nommée dans une cellule (d'une feuille séparée en général)

INDIC_ERREUR	
	0
1	Indicateur d'erreur
2	*erreur*

Cela permettra d'avoir toujours le même texte en cas d'erreur

2. Une colonne de comptage des erreurs a été ajoutée (c'est-à-dire du nombre de fois où on trouve INDIC_ERREUR dans la ligne, ici) :

E3		=NB.SI(A3:D3;"="&"*"&INDIC_ERREUR&"*")			
	A	B	C	D	E
1	Paie du mois de Octobre 2014 (avec gestion d'erreurs)				
2	Matricule	Nom	Salaire de base	Prime	erreurs ?
3	1000	*erreur*	*erreur*	300,00 €	2
4	1020	Inès Perret	1 600,00 €	200,00 €	0
5	1010	Jean Rigole	*erreur*fiche	100,00 €	1
6					0
7		totaux	calcul impossible	calcul impossible	3

3. et les formules ont évolué :
 - a. en B3

```
=SI(ESTVIDE(A3);"";
SI(ESTNA(RECHERCHEV(A3;TAB_PERSONNEL;2;FAUX));INDIC_ERREUR;
RECHERCHEV(A3;TAB_PERSONNEL;3;FAUX)&" "&RECHERCHEV(A3;TAB_PERSONNEL;2;FAUX)
)
)
```

- b. en C3

```
=SI(ESTVIDE(A3);"";
SI(ESTNA(RECHERCHEV(A3;TAB_PERSONNEL;1;FAUX));INDIC_ERREUR;
SI(ESTNA(RECHERCHEV(RECHERCHEV(A3;TAB_PERSONNEL;4;FAUX);TAB_EMPLOI;2;FAUX));INDIC_ERREUR&" sur fiche";
RECHERCHEV(RECHERCHEV(A3;TAB_PERSONNEL;4;FAUX);TAB_EMPLOI;2;FAUX)
)
)
)
```

- c. en C7 et E7

```
=SI(E7<>>0;"calcul impossible";SOMME(C3:C6))
```

```
=SI(E7<>>0;"calcul impossible";SOMME(D3:D6))
```

Exemple 2 : avec INDEX et EQUIV

L'objectif est ici de rechercher le tarif d'un vin d'une certaine année :

	A	B	C	D	E	F	G
1	Tarif des vins					rechercher un vin	
2	Annee					Vin	Gewurztraminer
3	Vin	2011	2012	2013		Année	2012
4	Sylvaner			4,50 €			
5	Riesling		10,80 €	6,00 €		Tarif	11,00 €
6	Gewurztraminer	20,50 €	11,00 €	7,00 €			
7	Pinot noir		9,50 €	7,00 €			

Si le vin n'a pas été trouvé dans la liste des vins (RECHERCHEV), message « ce vin n'existe pas » :

	A	B	C	D	E	F	G
1	Tarif des vins					rechercher un vin	
2	Annee					Vin	bordeaux
3	Vin	2011	2012	2013		Année	2012
4	Sylvaner		0,00 €	4,50 €			
5	Riesling	0,00 €	10,80 €	6,00 €		Tarif	ce vin n'existe pas
6	Gewurztraminer	20,50 €	11,00 €	7,00 €			
7	Pinot noir	0,00 €	9,50 €	7,00 €			

Si l'année ou le vin n'a pas été trouvé (INDEX et EQUIV), ou bien si le tarif n'est pas renseigné (ESTVIDE), message « NON DISPONIBLE » :

	A	B	C	D	E	F	G
1	Tarif des vins					rechercher un vin	
2	Annee					Vin	Gewurztraminer
3	Vin	2011	2012	2013		Année	2010
4	Sylvaner		0,00 €	4,50 €			
5	Riesling	0,00 €	10,80 €	6,00 €		Tarif	non disponible
6	Gewurztraminer	20,50 €	11,00 €	7,00 €			
7	Pinot noir	0,00 €	9,50 €	7,00 €			

Si le tarif est à 0.00, message « épuisé » (remarquer l'utilisation du caractère joker « * »)

	A	B	C	D	E	F	G
1	Tarif des vins					rechercher un vin	
2	Annee					Vin	Ries*
3	Vin	2011	2012	2013		Année	2011
4	Sylvaner		0,00 €	4,50 €			
5	Riesling	0,00 €	10,80 €	6,00 €		Tarif	epuisé
6	Gewurztraminer	20,50 €	11,00 €	7,00 €			
7	Pinot noir	0,00 €	9,50 €	7,00 €			

Sinon affichage du tarif :

	A	B	C	D	E	F	G
1	Tarif des vins					rechercher un vin	
2	Annee					Vin	Pinot*
3	Vin	2011	2012	2013		Année	2012
4	Sylvaner		0,00 €	4,50 €			
5	Riesling	0,00 €	10,80 €	6,00 €		Tarif	9,50 €
6	Gewurztraminer	20,50 €	11,00 €	7,00 €			
7	Pinot noir	0,00 €	9,50 €	7,00 €			

La formule est la suivante :

```
=SI(ESTNA(RECHERCHEV(G2;TAB_NOMVIN;1;FAUX));
"ce vin n'existe pas";
SI(OU(ESTNA(INDEX(TAB_TARIFVIN;EQUIV(G2;TAB_NOMVIN;0);EQUIV(G3;TAB_ANNEEVIN;0)));
ESTVIDE(INDEX(TAB_TARIFVIN;EQUIV(G2;TAB_NOMVIN;0);EQUIV(G3;TAB_ANNEEVIN;0)))
);
"non disponible";
SI(INDEX(TAB_TARIFVIN;EQUIV(G2;TAB_NOMVIN;0);EQUIV(G3;TAB_ANNEEVIN;0))=0;
"épuisé";
INDEX(TAB_TARIFVIN;EQUIV(G2;TAB_NOMVIN;0);EQUIV(G3;TAB_ANNEEVIN;0))
)
)
)
```

Ce genre de formule nécessite une structuration claire pour éviter les erreurs.

On constate la répétition d'une même recherche : une solution permettant de simplifier la formule serait de

- réserver une cellule (cachée de l'utilisateur, par exemple) qui effectuerait la recherche systématique du tarif

	H	I	J	K	L	M	N	O
1			Recherche du tarif					
2			9,50 €					

- puis utiliser directement cette cellule dans la formule initiale :

La formule devient alors :

```
=SI(ESTNA(RECHERCHEV(G2;TAB_NOMVIN;1;FAUX));
"ce vin n'existe pas";
SI(OU(ESTNA(RECH_TARIF);
ESTVIDE(RECH_TARIF)
);
"non disponible";
SI(RECH_TARIF=0;
"épuisé";
RECH_TARIF
)
)
)
```

```
=SI(ESTNA(RECHERCHEV(G2;TAB_NOMVIN;1;FAUX));"ce vin n'existe pas";
SI(OU(ESTNA(RECH_TARIF);ESTVIDE(RECH_TARIF));"non disponible"; SI(RECH_TARIF=0;"épuisé";RECH_TARIF)))
```